

# SAS Technical News

Autumn 2000

*For Higher  
Customer Satisfaction,  
We Bridge  
the SAS System  
Between  
Customer's World.*

## CONTENTS

- 1 特集 SAS V8eの新機能とV6環境からの切り替え
- 10 Q&A
- 15 日本SASユーザー会総会および研究発表会(SUGI-J 2000開催報告)
- 16 新刊マニュアルのご紹介
- 17 SASプログラマーズ・ワークショップ開催のお知らせ
- 18 Windows版SASシステム バージョン 8eリリースのお知らせ
- 18 SASトレーニングのお知らせ
- 19 最新リリース情報

## 特集

### SAS V8eの新機能と V6環境から の切り替え

#### はじめに

SAS V8e (Version 8.1 TS1M0 以後「V8e」と略します) がいよいよ出荷開始となりました。いままでSAS V6 (Version 6.07~6.12 以後「V6」と略します) をお使いいただいている皆様にとりまして、以下の3点は最も関心のある点だと思います。これらについて簡単にご紹介します。

1. 新機能及び拡張点は何か
2. 今までの資産は使えるのか
3. V6からV8eに切り替えるにはどうすれば良いか

1. SAS V8e SAS基本機能における新機能とV6からの拡張点  
V8eにおいては数々の新機能及び拡張点がありますが、ここではSAS基本機能を中心に「DATAステップ」「SASデータライブラリ」「出力」「他社製品とのデータ交換」の4点をご紹介します。

#### 1.1 DATAステップに関する新機能と拡張点

##### 【変数名】

変数名の長さが変更されたのは大きな拡張点の1つです。変更された項目とその長さは以下のとおりです。

| 項目  | V6での長さ | V8での長さ    |
|---|--------|-----------|
| <ul style="list-style-type: none"> <li>・変数名(Arrayステートメントで指定可能な配列含む)</li> <li>・SASデータセット名</li> <li>・SASカタログ名</li> <li>・マクロ名</li> <li>・マクロ変数名</li> <li>・ラベル付きステートメントの「ラベル」名</li> <li>・カタログエントリ名(SAS/AF SAS/FSP他)</li> </ul> | 8バイト   | 32バイト     |
| <ul style="list-style-type: none"> <li>・列ラベル</li> <li>・データセットラベル</li> </ul>   | 40バイト  | 256バイト    |
| <ul style="list-style-type: none"> <li>・1変数当たりの最大文字数</li> </ul>   | 200バイト | 32,767バイト |

「SASデータセットの世代管理」を行う場合、28バイトが最大になります。(世代管理については後述)

なお、[ライブラリ参照名][ファイル参照名][フォーマット名]および[インフォーマット名]についてはV6と同様に8バイトとなります。新しく追加された[VALIDVARNAME=]システムオプションを使用すると、V6同様に8バイト制限にできます。ただしV6互換の場合、8バイト長の変数名・データセット名を持つSASデータセットは操作が一切できなくなりますのでご注意ください。

#### 【SETステートメント】

SETステートメントでSASデータセットをOPENするタイミングを変更できるようになりました。

例)  
全データセットを同時にOPENする場合

```
DATA merge_data ;
SET table1 table2 table3 table4 table5 table6 table7 ;
RUN ;
```

上記はTABLE1～TABLE7を結合して、MERGE\_DATAというSASデータセットを作成する例です。V6ではTABLE1～TABLE7の全てを同時にOPENしていました。V8eでは下記のようにOPEN=DEFERを指定することでSASデータセットを順次OPENできます。これによってTABLEをOPENする際に消費されるメモリの使用量の低減がはかれます。下記の例はTABLE1をOPENし、読み終えてCLOSEした後、TABLE2以降を順次OPENします。

例)  
データセットを順次OPENする場合

```
DATA merge_data ;
SET table1 table2 table3 table4 table5 table6 table7 OPEN=DEFER ;
RUN ;
```

#### 【FILEステートメント】

今までINFILEステートメントにのみ存在していたDELIMITER(DLM)=オプション、DSDオプションがFILEステートメントにも追加されました。その結果、CSVファイルやタブ区切りファイルの出力を、簡単に行えるようになりました。

例)  
CSVファイルの出力

```
DATA _NULL_ ;
SET sasuser.class (OBS=3) ;
FILE PRINT DSD ;
PUT name sex age height weight ;
RUN ;
```

結果)

```
Alice,F,13,56.5,84
Becka,F,13,65.3,98
Gail,F,14,64.3,90
```

## 1.2 SASデータライブラリに係わる新機能と拡張点

### 【SASライブラリ・SASCatalogの参照】

ここではLIBNAMEステートメントのエンジン拡張および新しく追加されたCATNAMEステートメントを中心に紹介します。

#### ・LIBNAMEステートメント

##### 1) V8エンジンの指定追加

V8eではV6で作成されたSASデータセットをV6形式のまま参照、編集、追加ができます。SASシステムは、SASデータセットがV6形式・V8形式のどちらであるかを自動的に認識しますが、V6エンジン、V8エンジンの指定を行って認識させることもできます。

```
LIBNAME libref V8 'V8eのデータセット・カタログの格納ディレクトリ名';
LIBNAME libref V6 'V6のデータセット・カタログの格納ディレクトリ名';
```

これについては後述の「SAS V8eとV6資産の共存」で詳しく説明します。

##### 2) DBMSエンジンの指定追加

LIBNAMEステートメントを使用し、他社データベースのテーブルを参照できるようになりました。

例)  
Oracle内のテーブルを参照する場合

```
LIBNAME oralib oracle user=xxxx pw=yyyy path='dbmssrv' schema=educ ;
```

上記の場合はSAS/ACCESS to Oracleソフトウェアの契約が必要ですが、

3) ライブラリが割り当てられているSASライブラリの再割り当て  
SASDATA1が既に割り当てられているSASライブラリに対し、LIBNAMEステートメントのライブラリ参照名を使用して、更に別のライブラリ名を割り当てることができます。また、複数のライブラリ参照名を1つにまとめることができるようになりました。

例)  
ライブラリ参照名を使った再割り当て

```
LIBNAME sasdata 'd:\user\sampdat';
LIBNAME userdata (sasdata);
```

上記はライブラリ参照名SASDATAが割り当てられているSAMPDATに、さらに USERDATAという別のライブラリ参照名を割り当てている例です。

例)  
複数ライブラリ参照名の統合

```
LIBNAME v8data V8 'd:\user\samplib\v8';
LIBNAME v6data V6 'd:\user\samplib\v6';
LIBNAME sasdata (v8data v6data);
```

左記はV8DATAというV8形式のSASライブラリと、V6DATAというV6形式のSASライブラリを、SASDATAというライブラリ参照名で統合した例です。この状態でV6DATAのライブラリ割り当てを取り消しても、SASDATAはV8DATA、V6DATAの双方の内容を継続して参照することが可能です。

#### ・CATNAMEステートメント

CATNAMEステートメントは、散在している複数のSASカタログを1つのカタログ名で参照できる機能です。

例)

```
CATNAME work.catlib (sascat.catalog1 sascat.catalog2);
```

上記はカタログSASCAT.CATALOG1とSASCAT.CATALOG2を、WORK.CATLIBというカタログ参照名で参照することを実現しています。ただしSASCAT.CATALOG1に格納されたFRAMEエントリ等をWORK.CATLIBを通して編集することはできません。WORK.CATLIBは参照するための名前ではありません。また、ユーザ定義ライブラリに格納されたSASデータセットについては、LIBNAMEステートメントを使用せず、物理パス名で直接参照することが可能になりました。

例)

```
PROC PRINT DATA='D:¥資料¥sas77-¥宛¥class'; RUN;
```

上記はD:¥資料¥SAデータセットに格納されたCLASSというSASデータセット（物理ファイル名：class.sas7bdat または class.sd2）を参照する場合

#### 【世代管理】

V8eではGENMAと、GENNUMオプションを使用することで、SASデータセットの世代管理を実現しています。修正前のSASデータセットを参照する等の要望に応えることができます。

例)

DATALIB.CLASSを10世代まで管理する例

```
DATA datalib.class (GENMAX=10);
SET sasuser.class;
RUN;
```

GENMAXで何世代まで管理するかを指定します。

例)

3世代前のDATALIB.CLASSをプリントする例

```
PROC PRINT DATA=datalib.class (GENNUM=-3);
RUN;
```

GENNUMで処理を行うデータセットが何世代前のものかを指定します。

|        |                       |
|--------|-----------------------|
| GENNUM | 参照されるデータセット           |
| 0      | 最新のデータセット             |
| 正の整数   | 作成時から [ 世代目 ] のデータセット |
| 負の整数   | 最新から [ 世代前 ] のデータセット  |

世代管理を行う場合、SASシステムはデータセット名を内部的にDATALIB.CLASS#003等の名前で保持します。世代を示す番号4桁（バイト）が付加されるので、SASデータセット名の長さは28バイトまでとなります。

#### 【INDEX】

V8eではINDEXが作成されている変数をWHERE句で参照する処理が変更されました。

例)

SASデータセットMEMBERは変数IDについてINDEXが作成されています。

```
PROC PRINT DATA = member;
WHERE id ='15' OR id ='321' OR id ='215' OR id ='4' OR id ='510';
RUN;
```

これを実行した場合、V6とV8eでは以下のように出力結果が異なる場合があります。

| V6の実行結果（結果1） | V8eの実行結果（結果2） |
|--------------|---------------|
| ID           | ID            |
| '15'         | '4'           |
| '321'        | '15'          |
| '215'        | '215'         |
| '4'          | '321'         |
| '510'        | '510'         |

V8eの結果が異なる理由は以下のいずれかです。

1) オプションIDXWHERE=がNOであるため（デフォルトではYESになっています）

V8eではINDEXが作成された変数をWHERE句で取り扱う場合、INDEXを使用した処理を行うかどうかをIDXWHERE=オプションで決定することができます。

```
PROC PRINT DATA = member (IDXWHERE=YES);
WHERE id = '15' OR id = '321' OR id = '215' OR id = '4' OR id = '510';
RUN;
```

上記のように「IDXWHERE=YES」を指定することでV6と同様の結果を得ることができます（INDEXを使用した処理）。「IDXWHERE=NO」を指定するとINDEXを使用せず、常に順番にデータを読み込む処理を行います。

SQLプロシジャでは上記オプションは使用できません。

## 2) [ BUFNO= ] が適正な値ではないため

V8eではBUFNO=1がデフォルトで設定されていますが、この値が小さい場合、結果2のような場合があります。また、このような場合「OPTIONS MSGLEVEL=I ;」（デフォルトではNになっています）を指定すると、INDEXに関するメッセージをログに出力することができます。

```
INFO: Index ID not used. Sorting into index order may help.
INFO: Index ID not used. Increasing bufno to 3 may help.
```

これはINDEXを使用した処理をするためには、BUFNOを3まで引き上げる必要があることを示す情報です。「OPTIONS BUFNO=3 ;」を追加し、PRINT処理を実行するとV6と同様の処理結果を得られます。

## 【一貫性制約 (Integrity Constraints)】

SASデータセットにデータの追加・更新・削除を行う場合、そのデータ内容を一定のルールに従ってチェックし、ルールに違反した場合はその操作を無効とする機能が新規に追加されました。

### ・汎用制約 (General Constraints)

単独のデータセットの変数に対してあるルールを設定し、チェックを行う機能です。

例)

データセット [ EMPLOYEES ] を生成し、その中の1変数SALARYは4,000,000未満としないようにするために、SALARYが4,000,000未満のデータを排除する場合は、以下の書式で制約を設定します。

```
PROC SQL ;
CREATE TABLE employees
(idnum CHAR(5), name CHAR(30), salary NUM,
CONSTRAINT ok_salary CHECK(salary > 4000000));
```

上記はOK\_SALARYという制約名でルールが作成されています。このテーブルに対し、

```
PROC SQL ;
INSERT INTO employees
SET idnum='00001' name='山田' salary=3500000 ;
```

とSALARY=35000000のデータを追加しようとした場合、

```
ERROR: データセット WORK.EMPLOYEES への追加 / 更新
に失敗しました。 データ値が一貫性制約 OK_SALARY に従い
ません。
```

のメッセージを出力し、EMPLOYEESへの追加は行われません。

### ・参照制約 (Referential Constraints)

【主テーブル】に登録済みの場合にのみ、子テーブルにも登録可能とする等のチェックを行う機能です。

例)

データセットEMPLOYEESとSALESがあり、EMPLOYEESに登録されていないデータがSALESに登録されることを排除する場合、以下の様式で制約を設定します。

```
PROC SQL ;
CREATE TABLE employees
(idnum CHAR(5), name CHAR(30), salary NUM,
CONSTRAINT pk PRIMARY KEY (idnum))

CREATE TABLE sales
(idnum CHAR(5), name CHAR(30), age NUM,
CONSTRAINT fk FOREIGN KEY (idnum) REFERENCES employees
ON UPDATE RESTRICT ON DELETE RESTRICT);
```

上記はEMPLOYEESにIDNUMの主キーを設定し、SALESに外部キーIDNUM (EMPLOYEESの主キーIDNUMに依存) を設定しています。ここでSALESにEMPLOYEESへ登録されていないデータを追加しようすると、

```
ERROR: Observation was not added/updated because no
match was found for the foreign key value.
NOTE: テーブルを矛盾のない状態に戻すため上記のエラー
前の挿入を削除します。
```

のメッセージを出力し、SALESへの追加は行われません。

制約はSQLプロシジャ、DATASETSプロシジャ、SCL (SAS Component Language) でのみ生成・追加・削除が可能であり、DATAステップでは取り扱うことはできません。

## 【オーディット・トレイル (Audit Trails)】

オーディット・トレイルはSASデータセットに対して行われた操作をオーディット・トレイルファイル「(TYPE=AUDIT)」にログとして保持する機能です。

例)

データセットSASUSER.CLASSにオーディット・トレイルファイルを作成する例です。

```
PROC DATASETS LIB = sasuser ;
AUDIT class ;
INITIATE ; USER_VAR why $50 LABEL='異動理由' ;
RUN ;
```

上記のUSER\_VARステートメントは標準のオーディット・トレイル用の変数に任意の変数を追加します。SASUSER.CLASSに1件データを追加し、オーディット・トレイルファイルを表示すると以下の情報が出力されます。

(表示方法)

```
PROC PRINT DATA=sasuser.class(TYPE=AUDIT) ; RUN;
```

(上記の結果)

```

SAS システム          5
      2000年 8月10日 木曜日 15:34

NAME SEX AGE HEIGHT WEIGHT  why  _ATDATETIME_

Taro M  13 58.5  127 引越しによる転入 10AUG2000:17:03:32

_ATOBSNO_ _ATRETURNCODE_ _ATUSERID_ _ATOPCODE_ _ATMESSAGE_

20      .   jpnkcy   DA

```

\_ATで始まる変数は、オーデット・トレイルのために標準で提供されている変数です。

### 1.3 出力に係わる新機能と拡張点

#### 【ODS(Output Delivery System)】

ODSはSASにおける処理結果出力を革新的に変える機能です。ODSはOUTPUTウィンドウへの出力、HTMLファイルとしての出力、プリンタへの出力、SASデータセットとしての出力を行うことが可能です。ODS機能の一部についてご紹介します。

#### ・OUTPUTウィンドウへの出力 (Listing)

SAS V8eにおける基本的な利用方法です。OUTPUTウィンドウへの出力および外部ファイルへの出力を行う場合、この機能を使用します。

```

ODS LISTING (FILE='物理ファイル名' ; FILE参照名);
[出力を行いたい処理(例えば PROC PRINT DATA=sasuser.class ; RUN;)]
ODS LISTING CLOSE ;

```

外部ファイルへの出力を行う場合はFILE=を指定します。

「ODS LISTING CLOSE」の後、そのままOUTPUTウィンドウへの出力処理を行った場合、下記のメッセージが出て、何も出力されません。ODSステートメントによる制御には十分ご注意ください。

WARNING: 出力先がアクティブではありません。

#### ・HTMLファイルとしての出力

処理結果の出力をHTMLファイルとして生成・出力する機能です。

```

ODS HTML PATH='c:%temp' (URL=NONE) BODY='tabulate.html' ;
PROC TABULATE DATA=uridata ;
    CLASS shohin scd ; TABLES shohin,scd ;
RUN ;
ODS HTML CLOSE ;

```

上記は1プロシジャの処理結果を出力する例ですが、複数プロシジャの処理結果をFRAMEを使用したHTMLファイルとして生成することも可能です。

#### ・プリンタへの出力

処理結果の体裁や文字修飾を整えてプリンタに出力する機能です。

```

ODS PRINTER ;
PROC PRINT DATA=sasuser.class ;
RUN ;
ODS PRINTER CLOSE ;

```

上記はプリンタへ直接出力する例です。

#### ・SASデータセットとしての出力

処理結果をSASデータセットとして出力する機能です。V6ではデータセット出力ができなかったTABULATEプロシジャ等への使用が考えられます。

```

ODS OUTPUT REPORT=work.report ;
PROC TABULATE DATA=uridata ;
    CLASS shohin scd ; TABLES shohin,scd ;
RUN ;

```

ODS OUTPUTステートメントに続くREPORT=は使用プロシジャに応じて異なります。詳しくはオンラインヘルプ又は英語版マニュアル (The Complete Guide to the SAS Output Delivery System, Version 8 オーダーNo : 57241) をご参照ください。

ODSにつきましてはその他数多くの機能がありますので、詳しくはオンラインヘルプ又は英語版マニュアルをご参照ください。

### 1.4 他社製品とのデータ交換に係わる新機能と拡張点

#### 【IMPORT/EXPORTプロシジャ】

V6において評価版として提供されておりましたIMPORTプロシジャ/EXPORTプロシジャが、外部ファイルの入出力に使用するプロシジャとして利用できるようになりました。これらはカンマ区切り形式、タブ区切り形式ファイルの入出力に便利な機能を提供します。

例)

IMPORTプロシジャにてカンマ区切り形式ファイルの読み込みを行う場合

```

PROC IMPORT OUT= work.class
DATAFILE= "d:%temp%class.csv"
DBMS=CSV REPLACE;
GETNAMES=YES; /* 先頭行を変数名として使用する場合 */
Datarow=2; /* データ行は2行目より開始する場合 */
RUN;

```

例)

EXPORTプロシジャにてカンマ区切り形式ファイルの書き出しを行う場合

```

PROC EXPORT DATA= sasuser.class
OUTFILE= "d:%temp%class.csv"
DBMS=CSV REPLACE;
RUN;

```

更にSAS/ACCESS to PC File Formatsソフトウェアを契約している場合、IMPORT/EXPORTプロシジャでExcel97/2000形式のファイル(\*.XLS)やMS-ACCESS97/2000形式のファイル(\*.MDB)の読み込みや書き出しも可能となります。

例)

IMPORTプロシジャにてMS-ACCESS 2000テーブルの読み込みを行う場合

```
PROC IMPORT OUT= work.access
DATATABLE= "test1"
DBMS=ACCESS2000 REPLACE;
DATABASE="d:%temp%db1.mdb";
RUN;
```

例)

EXPORTプロシジャにてMS-ACCESS 2000テーブルへの書き出しを行う場合

```
PROC EXPORT DATA= sasuser.class
OUTTABLE= "class"
DBMS=ACCESS2000 REPLACE;
DATABASE="d:%temp%db1.mdb";
RUN;
```

V6の読み込みウィザードや書き込みウィザードではACCESSプロシジャを使用していましたが、V8eでは上記のIMPORT/EXPORTプロシジャを使用するように変更されました。

#### 【SAS/ACCESS LIBNAMEステートメント】

「1.2 SASデータライブラリに係わる新機能と拡張点」のLIBNAMEステートメントで記述した通り、LIBNAMEステートメントにエンジン指定することで、他社製品のデータを参照できる機能が追加されました。

## 2. SAS V8eとV6資産の共存

V6からV8eへの切替えを行う前に、双方のバージョンでの検証作業をする方々のために、資産の共存についてご紹介します。V8eではV6で作成したデータセット及びカタログの参照、データセットについては追加・更新・削除が行えるので、V6の資産をそのまま活用することができます。

### 2.1 同一コンピュータにV6とV8eを共存させる上での注意点

#### 【インストールのディレクトリについて】

双方のバージョンを共存させるためにV6とV8eのインストールディレクトリは完全に分けてください。V8eのディレクトリは主に以下の3つに分かれます。

- ・ [!SASROOT]ディレクトリ  
V6と同様にSASシステムの本体が格納されます。
- ・ SAS共有ディレクトリ  
ODSで使用するJavaアプレット、拡張エディタのモジュール等が格納されます。

- ・ SASUSERディレクトリ

SASUSERに格納される情報のためのディレクトリです。デフォルトは各ユーザーごとのProfilesディレクトリの直下になります。

#### 【拡張子の関連付けについて】

V6とV8eで主なファイルの拡張子が変わっています。

| ファイルの種類       | V6   | V8e       |
|---------------|------|-----------|
| LOG           | .LOG | .LOG      |
| OUTPUT        | .LST | .LST      |
| PROGRAM       | .SAS | .SAS      |
| ACCESSディスクリプタ | .SA2 | .SAS7BACS |
| SASカタログ       | .SC2 | .SAS7BCAT |
| SASデータセット     | .SD2 | .SAS7BDAT |
| テーブルインデックス    | .SI2 | .SAS7BNDX |
| データセットビュー     | .SV2 | .SAS7BVEW |
| Mddb          | .SM2 | .SAS7BMDB |
| ストアプログラム      | .SS2 | .SAS7BPGM |

ここで問題となるのはV6でしか使用されていない拡張子でも、V8eに関連付けが行われるということです。[\*.\*SAS]というファイルをクリックしてV6が起動する環境にするためには、ユーザーが関連付けを変更する必要があります。

#### 【その他】

「V8eのシステム必要条件」が適合しているコンピュータをご使用ください。以下の点についてご注意ください。

- ・ V8eではWindows95、Windows98、WindowsNTおよびWindows2000をサポートしています。ただし、Windows95、Windows98、WindowsNTについてはマイクロソフト社提供のY2K修正ファイルが適用済みであることをご確認ください。
- ・ V8eではヘルプファイルとしてHTML形式のヘルプを提供しています。これを利用するためにInternet Explorer 5を、SAS V8eインストール時にインストールすることができます。
- ・ SVGA (800x600) 以上のディスプレイが必要になります。

### 2.2 資産の共存について

#### 【LIBNAMEステートメント】

V6とV8eのデータセット・カタログはV8eから参照できますが、その管理方法はディレクトリ単位となります。よってV6のデータセット・カタログを格納するディレクトリとV8eのデータセット・カタログを格納するディレクトリは別々に分けていただく必要があります。このディレクトリ単位で「V6用ディレクトリ」と「V8e用ディレクトリ」とにLIBNAMEステートメントで定義します。

```
LIBNAME libref V6 'V6のデータセット・カタログの格納ディレクトリ名';
LIBNAME libref V8 'V8eのデータセット・カタログの格納ディレクトリ名';
```

V6、V8のエンジンを指定しなかった場合、V8eは以下のエンジンを自動的に割り当てます。

|                              | 割り当てエンジン |
|------------------------------|----------|
| ディレクトリにファイルが一切存在しない場合        | V8       |
| ディレクトリにV6のファイルのみ存在する場合       | V6       |
| ディレクトリにV8eのファイルのみ存在する場合      | V8       |
| ディレクトリにV6・V8eの両方のファイルが存在する場合 | V8       |

V6からはV8形式を参照できませんのでご注意ください。

例)  
V6でデータセットが認識できない操作

```
LIBNAME sasdata 'd:%sampdat'; /* D:%SAMPDATはファイルが一切存在しない */
DATA v8data;
  x = 'V8 DATA';
RUN;
```

上記の処理をV8eで行い、V6でSASDATA.V8DATAを参照しようとしても、SASデータセットV8DATAを認識することはできません。

例)  
V6でデータセットが認識できる操作

```
LIBNAME sasdata V6 'd:%sampdat';
DATA v6data;
  x = 'V6 DATA';
RUN;
```

上記をV8eで処理した場合（D:・SAMPDATにはファイルが存在しない）、SASDATAライブラリはV6形式になるので、SASデータセットV6DATAもV6形式で生成されます。

V6と共存させる場合、空のディレクトリをLIBNAMEステートメントでSASライブラリとして指定するには、この点に注意が必要です。

【V6ファイルの参照・編集について】

V6形式のファイルをV8eでどのように扱えるかは、以下のとおりです。

|                 |          |
|-----------------|----------|
| SASデータセット（テーブル） | 参照・追加・編集 |
| データセットビュー       | 参照       |
| SQL ビュー         | 参照       |
| SAS/ACCESS ビュー  | 参照・追加・編集 |
| SAS カタログ        | 参照       |

SAS/CONNECTソフトウェアを使用している場合、以下のようになります。

| サーバー | クライアント<br>エンジン | V6          | V8e         |             |
|------|----------------|-------------|-------------|-------------|
|      |                | V6          | V6          | V8          |
| V6   | V6             | U : V6形式    | U : V6形式    | U : V6形式    |
|      |                | D : V6形式    | D : V6形式    | D : V8形式    |
|      |                | RLS : 参・追・编 | RLS : 参・追・编 | RLS : 参・追・编 |
| V8e  | V6             | U : V6形式    | U : V6形式    | U : V6形式    |
|      |                | D : V6形式    | D : V6形式    | D : V8形式    |
|      |                | RLS : 参・追・编 | RLS : 参・追・编 | RLS : 参・追・编 |
|      | V8             | U : V8形式    | U : V8形式    | U : V8形式    |
|      |                | D : V6形式    | D : V6形式    | D : V8形式    |
|      |                | RLS : 参・追・编 | RLS : 参・追・编 | RLS : 参・追・编 |

凡例：

- [ U : V6形式 ] [ UPLOADプロシジャ ] で、[ サーバー ] にV6形式のSASデータセットを生成
- [ D : V6形式 ] [ DOWNLOADプロシジャ ] で、[ クライアント ] にV6形式のSASデータセットを生成
- [ RLS : 参・追・编 ] RLS経由でサーバーのSASデータセットの参照・追加・編集が可能      で囲まれているものは [ SASカタログ ] についても同操作が可能

例えば、[ サーバー : V8 (エンジンV6) ] [ クライアント : V8 (エンジンV8) ] は以下を表しています。

- U : V6形式 [ UPLOADプロシジャ ] で [ サーバー ] にV6形式SASデータセットを生成
- D : V8形式 [ DOWNLOADプロシジャ ] で [ クライアント ] にV8形式SASデータセット・カタログを生成
- RLS : 参・追・编 サーバーのV6データセットをクライアントよりRLS経由で参照・追加・編集可能

SAS/SHAREソフトウェアを使用している場合、以下のようになります。

| サーバー | クライアント<br>エンジン | V6          | V8e         |             |
|------|----------------|-------------|-------------|-------------|
|      |                | V6          | V6          | V8          |
| V6   | V6             | U : V6形式    | U : V6形式    | U : V6形式    |
|      |                | D : V6形式    | D : V6形式    | D : V8形式    |
|      |                | RLS : 参・追・编 | RLS : 参・追・编 | RLS : 参・追・编 |
| V8e  | V6             | U : V6形式    | U : V6形式    | U : V6形式    |
|      |                | D : V6形式    | D : V6形式    | D : V8形式    |
|      |                | RLS : 参・追・编 | RLS : 参・追・编 | RLS : 参・追・编 |
|      | V8             | U : V8形式    | U : V8形式    | U : V8形式    |
|      |                | D : V6形式    | D : V6形式    | D : V8形式    |
|      |                | RLS : 参・追・编 | RLS : 参・追・编 | RLS : 参・追・编 |

凡例：

[U : V6形式] [COPYプロシジャ]で、[サーバー]にV6形式のSASデータセットを生成

[D : V6形式] [COPYプロシジャ]で、[クライアント]にV6形式のSASデータセットを生成

[RLS : 参・追・编] サーバライブラリ経由でサーバのSASデータセットの参照・追加・編集が可能

■で囲まれているものは[SASカタログ]についても同操作が可能

例えば、[サーバー : V8 (エンジンV6)] [クライアント : V8 (エンジンV8)] は以下を表しています。

U : V6形式 [COPYプロシジャ]で[サーバー]にV6形式SASデータセットを生成

D : V8形式 [COPYプロシジャ]で[Client]にV6形式SASデータセット・カタログを生成

RLS : 参・追・编 サーバのV6データセットをクライアントよりサーバライブラリ経由で参照・追加・編集可能

なお、RLS (リモートライブラリサービス) 等でサーバのライブラリエンジンを指定する場合、以下の要領となります。

```
LIBNAME libref 'V6のSASライブラリのディレクトリ' SERVER=xxx
ENGINE=V6 ;
LIBNAME libref 'V8eのSASライブラリのディレクトリ' SERVER=xxx
ENGINE=V8 ;
```

### 2.3 互換性について

V6の資産 (データセット・カタログ) は原則としてV8eでも使用できます。またV6で稼動していたプログラムについても基本的にそのまま動作します。

ただし、SASシステムオプションによって動作が異なる場合がありますので、V8eを使用する際に一度オプション設定の内容をご確認ください。

### デフォルト値が変更されたシステムオプション (一般的なもの)

| オプション      | 値[ ( )はV6の値 ] | 考慮される影響   |
|------------|---------------|---|
| BUFNO      | 1(3)          | INDEXが生成されたデータセットを取り扱う際、処理効率が最適化されない可能性があります。   |
| YEARCUTOFF | 1920(1900)    | 西暦年下2桁のデータを日付値に変換する場合、その結果が異なる可能性があります。   |
| PROCLEAVE  | 8096(30K)     | 大容量のメモリを必要とするプロシジャに関しての設定値を見直す必要があります。  |
| SORTPGM    | BEST(SAS)     | SORTプロシジャに使用されるプログラムが「SAS」でない場合もあるため、処理速度等に影響が出る可能性があります。但し、LIBNAMEステートメントで他社製品のデータを参照している時は、「SAS」の指定の場合は十分な処理効率を得られない可能性があります。 |
| SYSLEAVE   | 8096(15K)     | V6において設定値を変更していた場合、見直していただく必要があります。   |

## 3. SAS V8eへの移行について

前項の、[2. SAS V8eとV6資産の共存]で説明しましたとおりV8eにてV6時代の資産を活用することは十分に可能ですが、V8eの新機能を利用するためには資産を移行する必要があります。ここではその移行方法についてご説明します。

### 3.1 移行に使用するプロシジャについて

メインフレーム又はVAX版SASシステムにおいて、V5からV6へSASデータライブラリを移行する時にはV5TOV6プロシジャという移行専用プロシジャを使用する必要がありましたが、V6からV8eへ移行するにはCOPYプロシジャを使ってV8形式のSASライブラリへ、自動的に変換を行えます。

手順としては以下のとおりです。

- LIBNAMEステートメントでV8エンジンを指定し、ライブラリ定義を行います。

```
LIBNAME v8lib V8 'd:\saslib\v8lib' ;
```

- COPYプロシジャを使用して、V8形式のSASライブラリにV6形式のデータセット等をコピーします。

```
LIBNAME v6lib V6 'd:\saslib\v6lib' ; /* V6形式のSASライブラリ */
PROC COPY IN=v6lib OUT=v8lib ;
RUN ;
```

- 上記で移行作業は終了です。

COPYプロシジャではデータセットの他にデータセットビュー、データセットインデックス等についても同様に変換が可能です。SASカタログについてもCOPYプロシジャでV8形式への変換は可能ですが、V6形式ではV8固有の機能が使用できませんので、完全なV8形式でお使いいただくためには、CPORTプロシジャ、CIMPORTプロシジャを使用した移送ファイル形式での移行をお奨めします。

## 1) V6にてCPORTプロシジャを使って移送ファイルを作成します。

```
FILENAME v6cat 'c:\temp\v6cat'; /* 移送ファイル */
LIBNAME v6lib V6 'd:\saslib\v6lib'; /* V6形式のSASライブラリ */
PROC CPORT LIB=v6lib FILE=v6cat MEMTYPE=CATALOG;
RUN;
```

## 2) V8eにてV8形式のSASライブラリ定義及びCIMPORTプロシジャによる移送ファイルの復元を行います。

```
LIBNAME v8lib V8 'd:\saslib\v8lib'; /* V8形式のSASライブラリ */
FILENAME v6cat 'c:\temp\v6cat'; /* 移送ファイル */
PROC CIMPORT LIB=v8lib INFILE=v6cat;
RUN;
```

## 3) 上記で移行作業は終了です。

## 3.2 移行用サンプルマクロについて

3.1で説明しました移行用のプロシジャについて簡単なマクロをご紹介します。このマクロを使用する際には、V6の資産が格納されているディレクトリとV8eの資産を格納するディレクトリを指定する必要があります。

例)

COPYプロシジャによる移行用マクロについて

```
%MACRO V6TOV8 (v6path,v8path);
LIBNAME v6lib V6 "&v6path";
LIBNAME v8lib V8 "&v8path";
PROC COPY IN=v6lib OUT=v8lib;
RUN;
%MEND;

%V6TOV8(V6資産のディレクトリ,V8e資産格納用ディレクトリ);
```

上記はV8eで実行してください。

実際の使用例は以下のとおりです。

```
%V6TOV8(d:\saslib\v6lib,d:\saslib\v8lib);
```

例)

CPORT / CIMPORTプロシジャによる移行用マクロについて

## 【V6で実行】

```
%MACRO V6TV8CAT (path);
LIBNAME v6lib "&path";
FILENAME v6cat "ltemp\v6cat";
PROC CPORT LIB=v6lib FILE=v6cat MEMTYPE=CATALOG;
RUN;
%MEND;

%V6TV8CAT(V6資産のディレクトリ);
```

上記のプログラムはTEMPディレクトリにV6CATというファイルを作成します。TEMPディレクトリはWindows上において環境変数で指定されているディレクトリです。

## 【V8eで実行】

```
%MACRO V6TV8CAT (path);
LIBNAME v8lib V8 "&path";
FILENAME v6cat "ltemp\v6cat";
PROC CIMPORT LIB=v8lib INFILE=v6cat;
RUN;
%MEND;

%V6TV8CAT(V8e資産格納用ディレクトリ);
```

TEMPディレクトリ内のV6CATファイルを使用します。

実際の使用例は以下の通りです。

```
/* V6における使用方法 */
%V6TV8CAT(d:\saslib\v6lib);

/* V8における使用方法 */
%V6TV8CAT(d:\saslib\v8lib);
```

必ず、V6とV8eそれぞれで実行してください。

これらのマクロは自動呼出マクロライブラリに登録すると、SASシステムを再度起動する度に「マクロプログラム」を実行する必要がないので便利です。マクロライブラリは既存のものを使用しても、新規に作成してもかまいません。マクロライブラリを新規に作成し、登録する方法は次の通りです。

- 適当なディレクトリ(例: C:\\$SAS\USR\SASMACRO)を作成し、マクロプログラムを記述したファイルを保存します。
- CONFIGファイル(例: C:\\$SAS\NLS\JA\SASV8.CFG(V8) C:\\$SAS\CONFIG.SAS(V6))内に「-SET SASAUTOS」と記述されている場所があります。その中に先ほど保存したディレクトリを記述してください。

```
-SET SASAUTOS (
    "c:\$sas\usr\sasmacro"
    .
    .
)
```

## おわりに

SAS V8e基本機能において、上記以外にも多数の新機能・拡張点があります。Windows版ではSASエクスプローラや拡張エディタなどインターフェイス部分に大きく変更されている箇所があります。この部分についても使い勝手の良い機能が用意されてありますので、V8e関連マニュアルおよびオンラインヘルプをご参照のうえご活用いただければ幸いです。

# Q&A

ダブルクォーテーション(")付きのデータより値のみを読み込む方法について  
 変数のラベルを全て削除する方法について  
 変数名の存在をチェックする方法について  
 全てのオブザベーションが欠損値である数値変数を削除する方法について  
 任意のオブザベーションを抽出する方法について  
 先頭に半角スペースを入力する方法について  
 メニューバー等を非表示にし、FRAMEを最大化して表示させる方法について  
 インストールしたプロダクト名を確認する方法について(UNIX版)  
 用量反応の非線形性を確認する方法について  
 行列の生成方法について  
 SASでLoess回帰を行う方法について

**Q** 値がダブルクォーテーション(")で閉じられている、カンマ区切りのデータがあるのですが、ダブルクォーテーション(")も読み込まれてしまいます。中の値のみを読み込みたいのですが、どのようにすればいいですか？

**A** INFILEステートメントに、DSDオプションを指定することにより、値のみを読み込むことができます。

例)

```
data work.test;
  infile cards dsd;
  input a b $;
  put a= b=;
cards;
"1234", "ABCD"
;
run;
```

**Q** あるデータセットから全ての変数のラベルを削除したいのですが、なにか良い方法がありますか？

**A** 通常、各変数に設定されているラベルを消去するためには例1のように変数名を指定してブランク(' ')を再設定する必要がありますが、例2のようにATTRIBステートメントを使用することで、全ての変数を表す \_ALL\_ を使用した記述方法が可能になります。

例1)

```
data sample;
  set sample;
  label var1= ' ' var2= ' ' ... ;
run;
```

例2)

```
data sample;
  set sample;
  attrib _all_ label= ' ';
run;
```

**Q** 指定する変数名が存在するかないかによって、異なった処理を行うことはできますか？

**A** バージョン6.11より新たに追加された%SYSFUNCマクロ関数を使用すると、DATAステップ内でいくつかのSCL関数が使用できるようになります。お使いのSASシステムのバージョンが6.11以上の場合、SCL関数のVARNUM関数を使用することで、変数の存在を簡単にチェックできます。VARNUM関数は、指定した変数の位置番号を返す関数です。指定した変数が存在しない場合、0を返します。

マクロプログラム例)

```
%macro varchk(ds,var);
  %if %sysfunc(varnum(%sysfunc(open(&ds)),&var))
  ne 0 %then %do;
  %put 変数 &var は存在します。;
  %end;
  %else %do;
  %put 変数 &var は存在しません。;
  %end;
%mend varchk;
```

また、CONTENTSプロシジャを使用すると、変数名を取得できます。お使いのバージョンが6.10以前の場合、このプロシジャを指定した変数名と照合することで、変数の存在を確認できます。

## マクロプログラム例)

```
%macro varchk(ds,var);
proc contents data=&ds out=temp (keep=name) noprint;
run;
%do;
  data _null_;
    retain exist "0";
    set temp ;
    if name=%upcase("&var") then exist="1";
    call symput("exist",exist);
run;
%if &exist=1 %then %do;
  %put 変数 &var は存在します。;
%end;
%else %do;
  %put 変数 &var は存在しません。;
%end;
%end;
%mend varchk;
```

## 実行例)

```
%varchk(データセット名,変数名);
```

# Q

全てのオブザベーションが欠損値である数値変数を削除する方法を教えてください。

# A

MEANSプロシジャには、欠損値でないオブザベーション数を計算するNという統計量キーワードがあります。この値を利用して、全てが欠損値である変数を求められます。

下記は、指定したデータセットから全てが欠損値である変数を削除するマクロプログラムの例です。

MEANSプロシジャのN統計量キーワードを別のデータセットに出力します。

後の処理に合うようにデータセットを転置します。

転置したデータセットより、目的の変数を抽出し、変数名、オブザベーション数をマクロ変数にセットします。

%IFステートメントにて、削除する変数が存在するかチェックします。存在する場合、SET ステートメントのDROPオプションを使用し、全てのオブザベーションが欠損値の変数を削除します。

## マクロプログラム例)

```
%macro dropvar(ds,out_ds);
  proc means data=&ds noprint;

  /* */
    output out=out(drop=_type_ _freq_) n=;
  run;

  /* */
    proc transpose data=out out=out2;
  run;

  /* */
    %let n=0;
    data _null_;
      set out2(where=(col1=0));
      call symput("var" || left(_n_),_name_);
      call symput("n",left(_n_));
    run;

  /* */
    data &out_ds;
      set &ds
        %if &n > 0 %then %do;
          (drop=
            %do i=1 %to &n;
              &&var&i
            %end;
          )
        %end;
    ;
  run;

%mend dropvar;
```

## 実行例)

```
%dropvar(入力データセット名,出力データセット名);
```

また、入力データセットdsと出力データセットout\_dsは同じデータセットでもかまいません。元のデータセットを残しておく場合は、別のデータセット名を指定してください。

**Q** トランザクションデータセットに基づいて、マスターデータセットから任意のオブザベーションを抽出したいのですが、どのようにすればよいでしょうか？

**A** MERGEステートメントのIN=オプションとBYステートメントによって、上記の操作は可能ですが、SETステートメントのPOINT=オプションとNOBS=オプションを併用すると、複数のデータセットのオブザベーションをそれぞれ独立に操作することができます。以下の例では、マスターデータセットのそれぞれのオブザベーションについて、トランザクションデータセットのオブザベーションをすべて読み込み、変数NAMEの値が一致するオブザベーションのみを外部ファイルに書き出しています。

例)

```
/* マスターデータセット */
data master;
  input name$ sex$ height weight;
cards;
Adam M 67 160
Arthur F 66 125
Barefoot M 68 158
Bostic F 74 170
;
run;

/* トランザクションデータセット */
data trans;
  input name$;
cards;
Adam
Barefoot
;
run;

data _null_;
  file 'c:\results.dat' mod;
  put 'NAME SEX HEIGHT WEIGHT';
  do i=1 to mn;
    set master point=i nobs=mn;
    do j=1 to tn;
      set trans (rename=(name=tname)) point=j nobs=tn;
      if name=tname then
        put name @11 sex @21 height @31 weight;
    end;
  end;

put;
put;
put;
stop;
run;
```

**Q** FSEDITプロシジャを使用し、データの入力を行っています。データの先頭に、半角スペースを入れたいのですが、左詰めされてしまいます。たとえば、「東京」ではなく「東京」のように半角スペースから始まるデータを入力する方法を教えてください。

**A** \$CHARw. インフォーマットを指定すると、先頭のブランクを有効にして入力することができます。wには、ブランクを含む、データの長さを入力します。

例)

```
proc fsedit data=fstest;
  informat a $char10.;
run;
```

**Q** Windows版SASシステムのSAS/AFソフトウェアを使用して、アプリケーションを構築しています。エンドユーザの操作ミスを防ぐために、FRAMEを呼び出す際、メニューバー、コマンドバー、ツールバーを表示せずに、かつFRAMEを最大化して表示したいのですが、何か良い方法はありますか？

**A** SASシステムを起動するアイコンのプロパティの設定と、Windows版SASシステムのSASシステムオプション、SASディスプレイマネージャコマンドにて、実現することができます。

下記の手順にて作業を行ってください。

1. SASシステムを起動するアイコンのプロパティ中にある[実行時の大きさ]を最大化に変更します。
2. [リンク先]に NOAWSMENU SASシステムオプションを追加し、メニューバーを非表示にします。

例)

```
C:\SAS\SAS.EXE -CONFIG C:\SAS\CONFIG.SAS -NOAWSMENU
```

3. [適用]、[OK]ボタンを選択します。
4. FRAMEのINITセクションにて、COMMAND、TOOLCLOSE、ZOOM SASディスプレイマネージャコマンドを実行し、コマンドバー、ツールバーを非表示に、そして、FRAMEを最大化します。

例)

```
init;
  call execcmd('command close; toolclose; zoom on');
return;
```

なお、上記のSASディスプレイマネージャコマンドを解除するには、下記のSASディスプレイマネージャコマンドを実行します。

例)

```
term;
  call execcmd('command bar; toolload; zoom off');
return;
```

Q

インストールしたプロダクト名を、SASを起動せずに確認する方法はありますか？

A

「!SASROOT/.install」ディレクトリ下(!SASROOTはインストールしたディレクトリ)に、SAS\_TOCという名前のファイルがあります。このファイルをエディタで開くと、インストールプログラム中に使用された情報を確認することができます。この中にあるPRODUCT\_LABELがインストールされたプロダクト名になります。また、「!SASROOT/.install/screens/」ディレクトリ下にインストール時の作業のログが残っていますので、インストール作業の手順を再確認したい場合等に、参照してください。

Q

用量反応の非線形性を確認するために、lack-of-fit検定を行いたいと思います。SASで実行する方法を教えてください。

A

lack-of-fit検定は、通常の分散分析の平方和から、直線回帰分析で説明できる平方和を引いて求めた平方和を利用します。SAS/STATソフトウェアのGLMプロシジャのTYPEの平方和を利用し、この平方和を求めることができます。

以下にプログラム例と出力結果を記載します。

実行手順

DATAステップで、分析に使用するデータセットを作成し、薬剤の用量を示す変数を2つ作成します。

(例 直線回帰分析用の変数DOSE・分散分析用の変数LACK)

GLMプロシジャのCLASSステートメントに分散分析用の変数を指定します。

GLMプロシジャのMODELステートメントに、応答変数、回帰分析用の変数、分散分析用の変数の順番で指定し、TYPE平方和を出力するオプションSS1を用います。

例)

```
*****データセットの作成 *****;
*薬剤の用量をDOSE・反応をRESPとしています。*;
data test1;
  input dose resp @@;
  lack=dose;                               /* */
cards;
1 123 2 148 3 159 4 166
1 128 2 146 3 161 4 165
1 120 2 142 3 154 4 171
1 137 2 147 3 159 4 123
;
run;

***** GLMプロシジャの実行*****;
proc glm data=test1;
  class lack;                               /* */
  model resp=dose lack/ss1;                 /* */
run;
```

出力結果)

| Source | DF | Type I SS     | Mean Square   | F Value | Pr > F |
|--------|----|---------------|---------------|---------|--------|
| DOSE   | 1  | 2010.01250000 | 2010.01250000 | 14.12   | 0.0027 |
| LACK   | 2  | 444.17500000  | 222.08750000  | 1.56    | 0.2499 |

\* LACKの部分で非直線性のlack-of-fit検定を行っています。

GLMプロシジャに関する詳細は、「SAS/STAT ソフトウェア：ユーザーズガイド、Version 6, First Edition」(注文番号1054 価格¥9,800)に記載されていますので、ご参照ください。

Q

SAS/IMLソフトウェアで、以下のようなデータから行列を簡単に構成する方法を教えてください。

例)

```
行 列   データ
1  3   1
3  6  -3
2  1   2
5  2  0.5
```



上のデータから下の行列を作る

```
. . 1 . . .
2 . . . .
. . . . -3
. . . . .
. . 5 . . .
```

**A**j関数は、全ての成分が同じである行列を生成します。次元縮小演算子を用いて、生成する行列の行数、および列数を取得し、成分が全て欠損値からなる行列を最初に作成して、それにデータの値を代入します。以下がプログラム例です。

例)

```
proc iml;

  x={1 3 1,
     3 6 -3,
     2 1 2,
     5 2 0.5}; /* データを行列として行列xに取り込みます */

  a=x[<>,1]; /* 生成する行列を求めます */
  b=x[<>,2];
  y=j(a,b,-); /* 求めた行数、列数からなる、成分が全て欠損値 */
               /* の行列yを作成します */

  do i=1 to nrow(x);
    /* nrow関数で行列xの行数を求め、その回数だけ */
    /* ループします */
    y[x[i,1],x[i,2]]=x[i,3];
    /* x[i,3]に入っている値のデータを新たに作る行列y */
    /* のx[i,1]行、x[i,2]列に代入します */
  end;
  print y; /* 行列yを表示します */
quit;
```

**Q** SASシステムを用いてLoess回帰を行うことはできますか？

**A** バージョン6.12のSASシステムでは、SAS/INSIGHTソフトウェアで説明変数が一つの場合のLoess回帰を実行できます。ただし、複数の説明変数があるものはサポートされていません。バージョン8eより、SAS/STATソフトウェアにLOESSプロシジャが追加されます。また、「ノンパラメトリック回帰」に分類される手法のうち、薄板スプライン(thin-plate spline)回帰を行うTPSPLINEプロシジャもバージョン8eのSAS/STATソフトウェアに追加されます。



# SUGI-J 2000

## 日本SASユーザー会総会 および研究発表会 (SUGI-J 2000開催報告)

8月31日、9月1日の2日間、東京全日空ホテルにて第19回日本SASユーザー会総会および研究発表会(SUGI-J2000)が開催されました。今回のSUGI-J 2000では、延べ参加人数2000名を超える盛況ぶりでした。これは、過去最高の記録であり、また、SASに対するお客様の関心の高さの表われであります。

発表された論文では、医薬をはじめシステム開発、金融、流通業界など多業種にわたり、今一番ホットなデータマイニングとSAS V8eで提供される協力的なツールであるODS(Output Delivery System)を中心に、総数62報の論文となりました。

初日のプレナリーセッションでは、日本SASユーザー会代表世話人である東京大学教授 大橋靖雄氏をはじめ、SUGI-J 2000年次総会チエアマンのキリンビール株式会社(日本SASユーザー会副代表世話人)本川裕氏、そして弊社株式会社SASインスティテュートジャパン代表取締役社長 デヴィッド C.フェンダーの挨拶がありました。本年も例年通り、日本SASユーザー会からは各優秀論文賞が発表されています。各賞は以下のとおりです。

### 最高論文賞

興和株式会社 菅波秀規氏、益田隆史氏  
「ダブルプログラミングによる統計解析の品質管理」

### 優秀論文賞

住友金属工業株式会社 水田匡彦氏  
「データマイニングによる製造プロセスの品質解析事例」

### 奨励賞

富士通イー・エム・ディ・セミコンダクタ株式会社 山本幸恵氏、渡部和浩氏  
「SASシステムによる半導体製造装置稼働データ分析ツールの構築」  
株式会社 CSK 柳澤彰子氏、松本雅彦氏  
「Webログを活用したデータマイニング」

### 企画賞

関西プロジェクト  
武田薬品工業株式会社 岩本 光司氏  
田辺製薬株式会社 北川 忠行氏  
日本ベーリンガーインゲルハイム株式会社 鍵村 達夫氏  
藤沢薬品工業株式会社 大津 洋氏、松岡 浄氏  
アストラゼネカ株式会社 伊藤 要二氏  
「V8のODS機能による総括報告書の電子化」

### ポスター賞

大学入試センター 菊池 賢一氏  
「Windows版SASとテキストエディタの連携パッケージ  
SASOLEの開発とその紹介」  
株式会社 竹中工務店 武藤 浩氏、建設省 小島 隆矢氏  
「オフィスの外観の『汚れ感』評価」

ポスター賞(グラフィック特別賞)  
キリンビール株式会社 長谷川 要氏

「SAS/GRAPH ソフトウェアを用いたフラクタル図形の作成」

また、日本SASユーザー会を支援する株式会社SASインスティテュートジャパンから、本年度より同社のパブリックリレーションズ活動に貢献したユーザ企業へ「SASインスティテュートジャパン特別賞」が制定され、SUGI-J 2000のプレナリーセッションの場を借り同賞の発表と授与式も行われています。第一回目の受賞企業は、以下のとおりです。

有限会社社企業行動デザイン研究所  
全日本空輸株式会社

米国SASインスティテュート本社からは、アジア太平洋ラテンアメリカ地域担当 副社長リー・リチャードソン氏が、日本では2年ぶりのメジャーバージョンアップとなるSASシステム バージョン8eのリリース、変わりつつあるeビジネス、成功するeビジネス等の報告がありました。

特別セッションも同時に開催され、米国SAS本社からR&Dディレクター ジョン・ブルックル バンク氏による Enterprise Minerソフトウェア バージョン4の「New Features in Enterprise Miner Release 4.0」と「Genomic Data Analysis with Enterprise Miner」のセッション、統計関連では、統計線形モデル研究開発グループ上級統計学研究員ラッセル・ウォルフインガー氏による「Statistical Enhancements for SAS Version 8e」のセッション、SASシステム バージョン8e関連では、SAS基本ソフトウェア研究開発部R&Dディレクターのステイブ・ビートルズ氏による「Version 8 特別セッション」等がありました。各セッションとも大勢のお客様が参加し、バージョン8eの期待と重要性の高さを感じました。

今回の論文発表の特徴として、バージョン8eの新機能の紹介はもちろんだ、様々な分野でのデータマイニングの事例が多く、もはやデータマイニングは特別な手法ではなく、通常の一プロセスとして、広範囲に広まってきていることを物語っています。

初日を締めくくる懇親会では、SUGI-J 2000の参加者をはじめ、日本SASユーザー会世話人会役員の方々、そして米国およびSASインスティテュートジャパンが一つになり、様々な意見交換を活発に行っていたのが非常に印象的でした。

来年のSUGI-J 2001では、SASシステム バージョン8eを中心に、強力なプロダクトとして、また、よりユーザフレンドリーで使いやすいツールや最新技術をSAS社からはご紹介していきたいと思えます。また来年、SUGI-J会場でお待ちしております。



# New Publications

## 新刊マニュアルのご紹介

SASより発売された新刊マニュアルをご紹介します。なお、こちらに記載されている価格はすべて税抜きです。ご購入については、専用のお申し込み用紙にてFAXで弊社「マニュアル販売係」までお申し込みください。

TEL : 03-3533-3835

FAX : 03-3533-3781

E-mail : booksale@jpn.sas.com

なお、最新のマニュアル案内は弊社ホームページ (<http://www.sas.com/japan/>)にて公開しておりますのであわせてご利用ください。

### Multiple Comparisons and Multiple Tests Using the SAS System

注文番号: 56648 (英語版)

価格 : 8,000円

本書は、最先端の方法、専門のマクロおよび証明された最適なプロシジャの使用法等について説明しています。

専門のマクロおよび多数の実例により、多数の推論を必要とするさまざまな問題のための解決を例証します。本書ではさらに、陥りやすい過ちや利点等についても説明し、それに関して、どれが目的に最も適切なものかを説明しています。製薬、工学、政府あるいは医学の研究者か科学者であれば、実データおよび実例にあったさまざまな方法を見つけることができます。本書には全体に渡って専門のプログラムコードやその説明が含まれており、一般線形の対比、多変量の手段の多重比較、混合モデル・離散データおよび生存データ解析の多数の推論も含まれています。

### Categorical Data Analysis Using the SAS System, Second Edition

注文番号 : 57998 (英語版)

価格 : 11,400円

統計学者や研究者向けの本書は、SASシステムを使用してカテゴリーカルデータ解析方法を行う際に非常に有用です。広範囲のアプリケーション例によって、いろいろな解析の中でFREQ、LOGISTIC、GENMODおよびCATMODプロシジャを学ぶことができます。他にはPHREGやNPAR1WAYプロシジャについても記載されています。また本書には分割表中の連関、テーブルの設定、ロジスティック回帰および条件付きロジスティック回帰、重み付き最小二乗法モデリング、反復測定解析、loglinearモデルおよび生物検定分析に関する記載されています。本書Second EditionはSASシステムバージョン8eのために改訂されました。追加された項目は、追加の正確なテスト、一般化された見積もりをしている方程式、LOGISTICプロシジャにおけるCLASSステートメントの使用、LOGISTICプロシジャを使用している正確なロジスティック回帰、母集団に対する特定のモデル、つまり平均的なモデルの比較等です。

### Getting Started with the SAS System, Version 8

注文番号 : 57551 (英語版)

価格 : 2,800円

本書を利用し、単純で図解入りの、ステップバイステップ形式の指示に従うことにより、基礎的なSASシステムの操作について学ぶことができます。データへのアクセスや管理、多様なレポートを使用しているデータの表現、統計方式を使ったデータ解析といった、SASシステムの環境についても学ぶことができます。SASシステムの新規ユーザおよびSASインタフェースの新しい特徴について知りたい人々にとって本書は最適です。

### Painless Windows : A Handbook for SAS User's, Second Edition

注文番号 : 56647 (英語版)

価格 : 7,100円

Windows環境の下でSASシステムはどう変わったでしょう。著者は、SASエクスプローラインタフェース、ODS (Output Delivery System)、ヘルプインタフェース、ウィンドウ結果、ウィンドウ機能、SASデータライブラリおよびファイルマネージメントなどの刺激的で新しい機能を紹介し、明確に解説しています。筆者はまたファイル編集、プログラムのサブミット、プリント、色・フォントなどのウィンドウ環境の調整、CONFIG.SAS・AUTOEXEC.SASといった起動時ファイルの編集、バッチモードの使用、DDE・OLEの使用、SAS/CONNECTソフトウェアの使用などさまざまなプロセスを紹介しています。本書は画面列、ステップバイステップ形式の説明、タスクが達成できるように簡潔な説明によって記載されています。Windows環境下でのすべてのSASプログラマーにとって本書は最適です。

### Professional SAS Programmers Pocket Reference, Third Edition

注文番号 : 58128 (英語版)

価格 : 3,600円

本書にはバージョン6以降で追加された多くの新しい特徴を含むSASシステムの構文、ルーチン、ファイル、オプション、コマンド等が記載されています。SASプログラマーに関する知識の有無にかかわらず、本書はSASプログラムを書く際に非常に役立ちます。

### The Complete Guide to the SAS Output Delivery System, Version 8

注文番号 : 57241 (英語版)

価格 : 7,800円

本書では、SASプロシジャの出力を、ブラウザで見ることができるようにする方法、高解像度ポストスクリプト・プリンタ用に生成する方法、SASのデータセットに変換する方法、自分のニーズにあったプロシジャ結果の出力結果に変更する方法、そしてSASがHTMLおよびプリンタ出力の中で使用する色、およびフォントを変更する方法などについて記載されています。また本書には上記の事柄以外にもSASプロシジャの結果や、DATAステッププログラムの結果についても記載されています。初級者でも、例題をコピーしたり修正することによりすばやくODSの基本機能を学習することができます。経験者には、オリジナルのSAS出力をTEMPLATEプロシジャを使用することによってSASの出力結果がどう変わるかを知ることができます。

The Little SAS Book: A Primer, Second Edition

注文番号：56649（英語版）

価格：5,500円

SASソフトウェアを学びたいがどこから始めたらいいのかわからない方のために、本書は読みやすくSASのもっとも一般的な特徴について解説しています。本書にはSASシステムに関する全般、SAS/STATソフトウェアへの入門、インポートプロシジャ/エクスポートプロシジャを含むSASデータセットへの取り込み方法などが記載されています。

本書には以下の情報が含まれています。DATAステップやPROCステップといったSASシステム、SASウィンドウ環境の使用法、永久的SAS データセットと一時的SAS データセットとの連携、データの出入力、データの更新、ブラウザで見るができるようにするためにHTMLファイルを生成することができるODS(Output Delivery System)の概要、マクロの概要、DATAステップデバッガーを含むSASプログラムのデバッグ、他のソフトウェア言語のプログラマーがSASプログラミング言語になれるための付録などがあります。

各トピックは例題および図解入りで使いやすい12ページレイアウトに分割されています。初級者でも本書は例題プログラムを使用しそれを確認する事ですばやくSASシステムを習得することができます。プログラム経験者には使い勝手の良いリファレンスとして使用することができます。

#### マニュアルキャンペーン2000のご案内

SUGI-J2000会場で行いました"マニュアルキャンペーン2000"を、ご好評につき、2000年10月31日まで延長いたします。今回は2000年9月にリリースされたSASシステムバージョン8eに対応したマニュアルおよびバージョン8eに対応した英語版トレーニングテキストを定価の15%引きで販売させていただきます。また、30,000円以上お買上げのお客様に「マニュアル割引券」をお渡しいたします。この機会をお見逃しなくご利用ください。

#### お問い合わせ先：マニュアル販売係

- TEL 03-3533-3835
- FAX 03-3533-3781
- E-mail booksale@jpn.sas.com

# SAS Programmers Workshop

## SASプログラマーズ・ワークショップ開催のお知らせ

本年9月に待望のWindows版SASシステムバージョン8eがリリースされました。このリリースに伴い、SASシステムバージョン8eに関するテクニカルなご紹介をいたします。内容につきましては、ODS機能(Output Delivery System)の紹介やSASシステムバージョン6との共存、また、クライアント・サーバ間にてバージョンV8eとバージョン6を使用した場合の注意点等をご説明します。日時、場所は、以下のとおりです。

#### 10月27日（金）

場所：株式会社SASインスティテュートジャパン 東京本社  
住所：東京都中央区勝どき1-13-1 イヌビル・カチドキ8F  
プレゼンテーションルーム

時間：14:00～17:00

定員：50名

#### 11月2日（木）

場所：株式会社SASインスティテュートジャパン 大阪支店  
住所：大阪市北区堂島浜1-4-16 アクア堂島西館12F  
プレゼンテーションルーム

時間：14:00～17:00

定員：50名

両日とも参加は無料となりますので、ぜひご参加ください。なお、定員に限りがございますので、お早めにお申し込み下さい。お申し込み方法は、「SASプログラマーズ・ワーク・ショップ参加申込用紙」をご記入の上、FAX、E-mailにてご送付ください。SASプログラマーズ・ワーク・ショップの申込用紙は下記のURLより入手できます。

<http://www.sas.com/japan/service/technical/pws.html>

その他、ご不明点、ご質問がございましたら、以下までご連絡下さい。

#### お問い合わせ先：マーケティング部

- TEL 03-3533-3780
- FAX 03-3533-1613
- E-mail saspws@jpn.sas.com

# Version 8e Release

## Windows版SASシステム バージョン 8eリリースのお知らせ

2000年8月29日、SASインスティテュート ジャパンはSASシステムの最新バージョンである「V8e」(Windows版)を、本年9月より順次出荷開始することを発表しました。

このSAS V8eは、これまでのSAS V6の機能がより強化されています。このSAS V8eにより、ODS (Output Delivery System) 機能の追加によるレポートの強化、SASデータライブラリ構造の変更によるデータアクセスの向上、その他これまで多くのSASシステムをご利用のお客様から熱望されておりました、SAS変数名のロングネームサポート等を実現しています。

また、現在様々な業界でご利用いただいている「Enterprise Minerソフトウェア」は、現行V2.0から大幅なアップグレードを実現したV4.0が、2000年11月にリリースされます。V4.0は、SAS V8e上で稼働し、より効率的なデータマイニングを実現できます。

今回リリース発表されたWindows版SAS V8eは、2001年夏には更なるアップグレードを予定しています。また、同時期に、他プラットフォームでもSAS V8eのリリースを予定しておりますので、ご期待ください。

### お申込み方法

V8eお申込み方法、お申込みにあたっての注意事項などは、下記URLをご参照ください。

- <http://www.sas.com/japan/service/v8/index.html>

### お問い合わせ先

- TEL 03-3533-6921 (東京本社)  
06-6345-5700 (大阪支店)  
093-512-5014 (九州営業所)
- E-mail [v8e@jpn.sas.com](mailto:v8e@jpn.sas.com)

# SAS Training

## SASトレーニングのお知らせ

### 新コースのご紹介!

#### 「データマイニング基礎」コース

「データマイニング基礎」コースが仲間入りしました。Enterprise Minerソフトウェアを使用してデータマイニングの定義をご説明します。「初級プログラミング」「SASによるデータ解析入門」を受講済の方ではじめてデータマイニングについて学習する方が対象になります。

1日間 ¥40,000 (チケット2ポイント)

東京: 10/20(金)、11/22(水)、12/20(水)

大阪: 11/27(月)

### 新サービスチケットのご紹介!

3チケット(受講3ポイント) ¥60,000

9/1よりお求めやすい「3チケット」を販売しています。詳細は「ホームページ」(<http://www.sas.com/japan>)トレーニングのご案内(10-12月)パンフレットで。

### 九州会場でトレーニング開始!

#### 「初級プログラミング」

2日間 ¥40,000 10/26(木)~27(金)

#### 「上級プログラミング」

2日間 ¥40,000 11/13(月)~14(火)

#### 「SASによるデータ解析入門」

2日間 ¥40,000 12/4(月)~5(火)

会場: 北九州市小倉北区浅野2-14-1 小倉興産KMMビル3F

SASインスティテュート ジャパン 九州営業所

(JR小倉駅北口より徒歩1分・西鉄バス小倉駅前下車徒歩3分)

### お問い合わせ先: トレーニング担当

- TEL 03-3533-3835
- FAX 03-3533-3781
- E-mail [training@jpn.sas.com](mailto:training@jpn.sas.com)

# Latest Releases

## 最新リリース情報

### PCプラットフォーム

|            |                  |       |
|------------|------------------|-------|
| Windows版   | SASシステムリリース 6.12 | TS060 |
| Windows版   | SASシステムリリース 8.1  | TS1MO |
| OS/2版      | SASシステムリリース 6.12 | TS020 |
| Macintosh版 | SASシステムリリース 6.12 | TS040 |

### ミニコンピュータプラットフォーム

|              |                  |       |
|--------------|------------------|-------|
| OpenVMS AXP版 | SASシステムリリース 6.12 | TS020 |
| OpenVMS VAX版 | SASシステムリリース 6.08 | TS407 |

### UNIXプラットフォーム

|                |                  |       |
|----------------|------------------|-------|
| MIPS ABI版      | SASシステムリリース 6.11 | TS040 |
| Digital Unix版  | SASシステムリリース 6.12 | TS040 |
| SunOS/Solaris版 | SASシステムリリース 6.12 | TS060 |
| HP-UX版         | SASシステムリリース 6.12 | TS060 |
| AIX版           | SASシステムリリース 6.12 | TS060 |
| ABI+版          | SASシステムリリース 6.11 | TS040 |

### メインフレームプラットフォーム

|       |                   |       |
|-------|-------------------|-------|
| MVS版  | SASシステムリリース 6.09E | TS470 |
| MSP版  | SASシステムリリース 6.09E | TS470 |
| VOS3版 | SASシステムリリース 6.09E | TS470 |
| CMS版  | SASシステムリリース 6.08  | TS410 |

## SAS Technical News Autumn 2000

発行  
株式会社SASインスティテュートジャパン

テクニカルニュースに関するお問い合わせ先

テクニカルサポートグループ  
TEL: 03-3533-3877  
FAX: 03-3533-3781  
E-mail: technews@jpn.sas.com



東京本社  
〒104-0054  
東京都中央区勝どき1-13-1  
イヌイビル・カチドキ 8F  
Tel 03 (3533) 6921  
Fax 03 (3533) 6927

大阪支店  
〒530-0004  
大阪市北区堂島浜1-4-16  
アクア堂島西館 12F  
Tel 06 (6345) 5700  
Fax 06 (6345) 5655

九州営業所  
〒802-0001  
北九州市小倉北区浅野2-14-1  
小倉興産KMMビル 3F  
Tel 093 (512) 5014  
Fax 093 (512) 5016

株式会社SASインスティテュートジャパン [www.sas.com/japan/](http://www.sas.com/japan/)